Using Stata for data management and reproducible research

Christopher F Baum

Boston College and DIW Berlin

NCER, Queensland University of Technology, March 2014
Stata is a full-featured statistical programming language for Windows, Mac OS X, Unix and Linux. It can be considered a “stat package,” like SAS, SPSS, RATS, or eViews.

Stata is available in several versions: Stata/IC (the standard version), Stata/SE (an extended version) and Stata/MP (for multiprocessing). The major difference between the versions is the number of variables allowed in memory, which is limited to 2,047 in standard Stata/IC, but can be up to 32,767 in Stata/SE or Stata/MP. The number of observations in any version is limited only by your computer’s memory.
Stata/SE relaxes the Stata/IC constraint on the number of variables, while Stata/MP is the multiprocessor version, capable of utilizing 2, 4, 8, 16... processors available on a single computer. Stata/IC will meet most users’ needs; if you have access to Stata/SE or Stata/MP, you can use that program to create a subset of a large survey dataset with fewer than 2,047 variables. Stata runs on all 64-bit operating systems, and can access larger datasets on a 64-bit OS, which can address a larger memory space.

All versions of Stata provide the full set of features and commands: there are no special add-ons or ‘toolboxes’. Each copy of Stata includes a complete set of manuals (over 11,000 pages) in PDF format, hyperlinked to the on-line help.
A Stata license may be used on any machine which supports Stata (Mac OS X, Windows, Linux): there are no machine-specific licenses for Stata version 11 onward. You may install Stata on a home and office machine, as long as they are not used concurrently. Licenses can be either annual or perpetual.

Stata works differently than some other packages in requiring that the entire dataset to be analyzed must reside in memory. This brings a considerable speed advantage, but implies that you may need more RAM (memory) on your computer. There are 32-bit and 64-bit versions of Stata, with the major difference being the amount of memory that the operating system can allocate to Stata (or any other application).
In some cases, the memory requirement may be of little concern. Stata is capable of holding data very efficiently, and even a quite sizable dataset (e.g., more than one million observations on 20–30 variables) may only require 500 Mb or so. You should take advantage of the `compress` command, which will check to see whether each variable may be held in fewer bytes than its current allocation.

For instance, indicator (dummy) variables and categorical variables with fewer than 100 levels can be held in a single byte, and integers less than 32,000 can be held in two bytes: see `help datatypes` for details. By default, floating-point numbers are held in four bytes, providing about seven digits of accuracy. Some other statistical programs routinely use eight bytes to store all numeric variables.
The memory available to Stata may be considerably less than the amount of RAM installed on your computer. If you have a 32-bit operating system, it does not matter that you might have 4 Gb or more of RAM installed; Stata will only be able to access about 1 Gb, depending on other processes’ demands.

To make most effective use of Stata with large datasets, you should use a computer with a 64-bit operating system. Stata will automatically install a 64-bit version of the program if it is supported by the operating system. All Linux, Unix and Mac OS X computers today come with 64-bit operating systems.
Stata is eminently portable, and its developers are committed to cross-platform compatibility. Stata runs the same way on Windows, Mac OS X, Unix, and Linux systems. The only platform-specific aspects of using Stata are those related to native operating system commands: e.g. is the file to be accessed

C:\Stata\StataData\myfile.dta
or
/users/baum/statadata/myfile.dta

Perhaps unique among statistical packages, Stata’s binary data files may be freely copied from one platform to any other, or even accessed over the Internet from any machine that runs Stata. You may store Stata’s binary datafiles on a webserver (HTTP server) and open them on any machine with access to that server.
Stata has traditionally been a command-line-driven package that operates in a graphical (windowed) environment. Stata version 11 (released June 2009), version 12 (released July 2011) and version 13 (released June 2013) contain a graphical user interface (GUI) for command entry via menus and dialogs. Stata may also be used in a command-line environment on a shared system (e.g., a Unix server) if you do not have a graphical interface to that system.

A major advantage of Stata’s GUI system is that you always have the option of reviewing the command that has been entered in Stata’s Review panel. Thus, you may examine the syntax, revise it in the Command panel and resubmit it. You may find that this is a more efficient way of using the program than relying wholly on dialogs.
Stata version 13, default screen appearance:
The Toolbar contains icons that allow you to Open and Save files, Print results, control Logs, and manipulate windows. Some very important tools allow you to open the Do-File Editor, the Data Editor and the Data Browser.

The Data Editor and Data Browser present you with a spreadsheet-like view of the data, no matter how large your dataset may be. The Do-File editor, as we will discuss, allows you to construct a file of Stata commands, or “do-file”, and execute it in whole or in part from the editor.
The foot of the screen also contains an important piece of information: the Current Working Directory, or $cwd$. In the screenshot, it is listed as `cfbaum > Documents >`. The $cwd$ is the directory to which any files created in your Stata session will be saved. Likewise, if you try to open a file and give its name alone, it is assumed to reside in the $cwd$. If it is in another location, you must change the $cwd$ [File→Change Working Directory] or qualify its name with the directory in which it resides.

You generally will not want to locate or save files in the default $cwd$. A common strategy is to set up a directory for each project or task in a convenient location in the filesystem and change the $cwd$ to that directory when working on that task. This can be automated in a do-file with the `cd` command.
There are several panels in the default interface: the Review, Results, Command, Variables and Properties panels. You may alter the appearance of any panel in the GUI using the Preferences → General dialog, and make those changes on a temporary or permanent basis.

As you might expect, you may type commands in the Command panel. You may only enter one command in that panel, so you should not try pasting a list of several commands. When a command is executed—with or without error—it appears in the Review panel, and the results of the command (or an error message) appears in the Results panel. You may click on any command in the Review panel and it will reappear in the Command panel, where it may be edited and resubmitted.
Once you have loaded data into the program, the Variables panel will be populated with information on each variable, as you can see in the example. That information includes the variable name, its label (if any), its type and its format. This is a subset of information available from the `describe` command.

Let’s look at the interface after I have loaded one of the datasets provided with Stata, `uslifeexp`, with the `sysuse` command and given the `describe` and `summarize` commands:
Overview of the Stata environment

Stata's user interface

1. `sysuse uslifeexp`
2. `describe`
3. `summarize`

```
. summarize

Variable | Obs | Mean | Std. Dev. | Min | Max
---------|-----|------|-----------|-----|-----
  year    | 100 | 1949.5 | 29.81149  | 1900 | 1999
   le     | 100 | 64.829 | 9.158628  | 39.1 | 76.7
  le_male | 100 | 62.302 | 8.436369  | 36.6 | 73.9
  le_female| 100 | 67.51  | 9.834987  | 42.2 | 79.5
    le_w  | 100 | 65.608 | 9.171269  | 39.8 | 77.3
  le_male | 100 | 63.143 | 8.503954  | 37.1 | 74.6
  le_wfemale| 100 | 68.434 | 9.797167  | 43.2 | 80
    le_b  | 100 | 56.933 | 12.48937  | 30.8 | 71.4
  le_bmale| 100 | 53.589 | 11.4569   | 29.1 | 67.8
 le_bfemale| 100 | 58.567 | 13.5409   | 32.5 | 74.8
```
Notice that the three commands are listed in the Review panel. If any had failed, the \(_rc\) column would contain a nonzero number, in red, indicating the error code.

The Variables panel contains the list of variables and their labels.

The Results panel shows the effects of `summarize`: for each variable, the number of observations, their mean, standard deviation, minimum and maximum. If there were any string variables in the dataset, they would be listed as having zero observations.
Try it out: type the commands

sysuse uslifeexp
describe
summarize

Take note of an important design feature of Stata. If you do not say what to describe or summarize, Stata assumes you want to perform those commands for every variable in memory, as shown here. As we shall see, this design principle holds throughout the program.
We may also write a do-file in the do-file editor and execute it. The Do-File Editor icon on the Toolbar brings up a window in which we may type those same three commands, as well as a few more:

```
sysuse uslifeexp
describe
summarize
notes
// average life expectancy, 1900–1949
summarize le if year < 1950
// average life expectancy, 1950–1999
summarize le if year >= 1950
```

After typing those commands into the window, the rightmost icon, with tooltip Do, may be used to execute them.
Overview of the Stata environment

Using Stata

Christopher F Baum (BC / DIW)  Using Stata  NCER/QUT, 2014 18 / 138
In this do-file, I have included the `notes` command to display the notes saved with the dataset, and included two comment lines. There are several styles of comments available. In this style, anything on a line following a double slash (`//`) is ignored. You may also place an asterisk (`*`) on the left margin to indicate a comment, or surround several comment lines in a do-file with the `/ * . . .  */` notation.

If a command is too long to fit comfortably on a single line, you may continue it on successive lines by placing a triple slash (`///`) at the end of each line.

You may use the other icons in the Do-File Editor window to save your do-file (to the `cwd` or elsewhere), print it, or edit its contents. You may also select a portion of the file with the mouse and execute only those commands.
Overview of the Stata environment

Using the Do-File Editor

Christopher F Baum (BC / DIW)  
Using Stata  
NCER/QUT, 2014  
20 / 138
**Try it out:** use the Do-File Editor to save and reopen the do-file `S1.1.do`, and run the file.

Try selecting only those last four lines and run those commands.
The rightmost menu on the menu bar is labeled Help. From that menu, you can search for help on any command or feature. The Help Browser, which opens in a Viewer window, provides hyperlinks, in blue, to additional help pages. At the foot of each help screen, there are hyperlinks to the full manuals, which are accessible in PDF format. The links will take you directly to the appropriate page of the manual.

You may also search for help at the command line with `help command`. But what if you don’t know the exact command name? Then you may use the `search` command, which may be followed by one or several words.
Results from `search` are presented in a Viewer window. Those commands will present results from a keyword database and from the Internet: for instance, FAQs from the Stata website, articles in the *Stata Journal* and *Stata Technical Bulletin*, and downloadable routines from the SSC Archive (about which more later) and user sites.

**Try it out:** when you are connected to the Internet, type the commands

```
search baum, au
search baum
```

Note the hyperlinks that appear on URLs for the books and journal articles, and on the individual software packages (e.g., `st0030_3`, `archlm`).
One of Stata’s great strengths is that it can be updated over the Internet. Stata is actually a web browser, so it may contact Stata’s web server and enquire whether there are more recent versions of either Stata’s executable (the kernel) or the ado-files. This enables Stata’s developers to distribute bug fixes, enhancements to existing commands, and even entirely new commands during the lifetime of a given major release (including ‘dot-releases’ such as Stata 12.1).

Updates during the life of the version you own are free. You need only have a licensed copy of Stata and access to the Internet (which may be by proxy server) to check for and, if desired, download the updates.
Another advantage of the command-line driven environment involves extensibility: the continual expansion of Stata’s capabilities. A command, to Stata, is a verb instructing the program to perform some action.

Commands may be “built in” commands—those elements so frequently used that they have been coded into the “Stata kernel.” A relatively small fraction of the total number of official Stata commands are built in, but they are used very heavily.
The vast majority of Stata commands are written in Stata’s own programming language—the “ado-file” language. If a command is not built in to the Stata kernel, Stata searches for it along the `adopath`. Like the `PATH` in Unix, Linux or DOS, the `adopath` indicates the several directories in which an ado-file might be located. This implies that the “official” Stata commands are not limited to those coded into the kernel.

**Try it out:** give the

```
adopath
```

command in Stata.
If Stata’s developers tomorrow wrote a new command named “foobar”, they would make two files available on their web site: `foobar.ado` (the ado-file code) and `foobar.sthlp` (the associated help file). Both are ordinary, readable ASCII text files. These files should be produced in a text editor, not a word processing program.
The importance of this program design goes far beyond the limits of official Stata. Since the adopath includes both Stata directories and other directories on your hard disk (or on a server’s filesystem), you may acquire new Stata commands from a number of web sites. The *Stata Journal (SJ)*, a quarterly peer-reviewed journal, is the primary method for distributing user contributions. Between 1991 and 2001, the *Stata Technical Bulletin* played this role, and a complete set of issues of the *STB* are available on line at the Stata website.
The *SJ* is a subscription publication (articles more than three years old freely downloadable), but the *ado-* and *sthlp-*files may be freely downloaded from Stata’s web site. The Stata *help* command accesses help on all installed commands; the Stata *search* command will locate commands that have been documented in the *STB* and the *SJ*, and with one click you may install them in your version of Stata. Help for these commands will then be available in your own copy.
User extensibility: the SSC archive

But this is only the beginning. Stata users worldwide participate in the StataList listserv, and when a user has written and documented a new general-purpose command to extend Stata functionality, they announce it on the StataList listserv (to which you may freely subscribe: see Stata’s web site).

Since September 1997, all items posted to StataList (over 1,500) have been placed in the Boston College Statistical Software Components (SSC) Archive in RePEc (Research Papers in Economics), available from IDEAS (http://ideas.repec.org) and EconPapers (http://econpapers.repec.org).
Any component in the SSC archive may be readily inspected with a web browser, using IDEAS’ or EconPapers’ search functions, and if desired you may install it with one command from the archive from within Stata. For instance, if you know there is a module in the archive named `mvsumm`, you could use `ssc describe mvsumm` to learn more about it, and `ssc install mvsumm` to install it if you wish. Anything in the archive can be accessed via Stata’s `ssc` command: thus `ssc describe mvsumm` will locate this module, and make it possible to install it with one click.

Windows users should not attempt to download the materials from a web browser; it won’t work.
Try it out: when you are connected to the Internet, type

```stata
ssc describe mvsumm
ssc install mvsumm
help mvsumm
```
The command `ssc new` lists, in the Stata Viewer, all SSC packages that have been added or modified in the last month. You may click on their names for full details. The command `ssc hot` reports on the most popular packages on the SSC Archive.

The Stata command `adoupdate` checks to see whether all packages you have downloaded and installed from the SSC archive, the *Stata Journal*, or other user-maintained net from... sites are up to date. `adoupdate` alone will provide a list of packages that have been updated. You may then use `adoupdate, update` to refresh your copies of those packages, or specify which packages are to be updated.
The importance of all this is that Stata is *infinitely extensible*. Any ado-file on your `adopath` is a full-fledged Stata command. Stata’s capabilities thus extend far beyond the official, supported features described in the Stata manual to a vast array of additional tools.

As the current directory is on the `adopath`, use the Do-File Editor to an ado-file in that directory named `hello.ado`. 

**Try it out:**

```plaintext
program define hello
display "Stata says hello!"
end
exit
```

Stata will now respond to the command `hello`. It’s that easy.
Let us consider the form of Stata commands. One of Stata’s great strengths, compared with many statistical packages, is that its command syntax follows strict rules: in grammatical terms, there are no irregular verbs. This implies that when you have learned the way a few key commands work, you will be able to use many more without extensive study of the manual or even on-line help.
The fundamental syntax of all Stata commands follows a *template*. Not all elements of the template are used by all commands, and some elements are only valid for certain commands. But where an element appears, it will appear in the same place, following the same grammar.

Like Unix or Linux, Stata is case sensitive. Commands must be given in lower case. For best results, keep all variable names in lower case to avoid confusion.
The general syntax of a Stata command is:

```
[prefix_cmd:] cmdname [varlist] [=exp]
    [if exp] [in range]
    [weight] [using...] [,options]
```

where elements in square brackets are optional for some commands.

In some cases, only the `cmdname` itself is required. `describe` without arguments gives a description of the current contents of memory (including the identifier and timestamp of the current dataset), while `summarize` without arguments provides summary statistics for all (numeric) variables. Both may be given with a `varlist` specifying the variables to be considered.

What are the other elements?
The varlist

`varlist` is a list of one or more variables on which the command is to operate: the subject(s) of the verb. Stata works on the concept of a single set of variables currently defined and contained in memory, each of which has a name. As the `describe` command will show you, each variable has a data type (various sorts of integers and reals, and string variables of a specified maximum length). The varlist specifies which of the defined variables are to be used in the command.
The order of variables in the dataset matters, since you can use hyphenated lists to include all variables between first and last. (The `order` and `move` commands can alter the order of variables.) You can also use “wildcards” to refer to all variables with a certain prefix. If you have variables `pop60`, `pop70`, `pop80`, `pop90`, you can refer to them in a `varlist` as `pop*` or `pop?0`.
The exp clause is used in commands such as `generate` and `replace` where an algebraic expression is used to produce a new (or updated) variable. In algebraic expressions, the operators `==`, `&`, `|` and `!` are used as equal, AND, OR and NOT, respectively. The `\` operator is used to denote exponentiation. The `+` operator is overloaded to denote concatenation of character strings.
The if and in clauses

Stata differs from several common programs in that Stata commands will automatically apply to all observations currently defined. You need not write explicit loops over the observations. You can, but it is usually bad programming practice to do so. Of course you may want not to refer to all observations, but to pick out those that satisfy some criterion. This is the purpose of the \texttt{if exp} and \texttt{in range} clauses. For instance, we might:

\begin{verbatim}
sort price
list make price in 1/5
\end{verbatim}

to determine the five cheapest cars in auto.dta. The 1/5 is a \texttt{numlist}: in this case, a list of observation numbers. $l$ is the last observation, thus \begin{verbatim}
list make price in -5/$l$
\end{verbatim}
will list the five most expensive cars in auto.dta.
Even more commonly, you may employ the *if exp* clause. This restricts the set of observations to those for which the “exp”, a Boolean expression, evaluates to true. Stata’s missing value codes are greater than the largest positive number, so that the last command would avoid listing cars for which the price is missing.

```
list make price if foreign==1
```

or

```
list make price if foreign
```

lists only foreign cars, and

```
list make price if price > 10000 & !mi(price).
```

lists only expensive cars (in 1978 prices). Note the double equal in the *exp*. A single equal sign, as in the C language, is used for assignment; double equal for comparison.
The using clause

Some commands access files: reading data from external files, or writing to files. These commands contain a *using* clause, in which the filename appears. If a file is being written, you must specify the “replace” option to overwrite an existing file of that name.

Stata’s own binary file format, the `.dta` file, is cross-platform compatible, even between machines with different byte orderings (low-endian and high-endian). A `.dta` file may be moved from one computer to another using `ftp` (in binary transfer mode).
To bring the contents of an existing Stata file into memory, the command:

```
use file [,clear]
```

is employed (clear will empty the current contents of memory). You must have sufficient memory for Stata to load the entire file, as Stata’s speed is largely derived from holding the entire data set in memory.
Reading and writing binary (.dta) files is much faster than dealing with text (ASCII) files (with the `insheet` or `infile` commands), and permits variable labels, value labels, and other characteristics of the file to be saved along with the file. To write a Stata binary file, the command

```
save file [,replace]
```

is employed. The `compress` command can be used to economize on the disk space (and memory) required to store variables.

Stata’s version 10, 11 and 12 datasets cannot be read by version 8 or 9; to create a compatible dataset, use the `saveold` command. Likewise, Stata 13 uses a new dataset format to accommodate long string variables. `saveold` in Stata 13 will create a dataset usable (except for long strings, or `strLs`) in version 11 or 12.
Many commands make use of options (such as `clear on use`, or `replace on save`). All options are given following a single comma, and may be given in any order. Options, like commands, may generally be abbreviated (with the notable exception of `replace`).
Programmability of tasks

Stata may be used in an interactive mode, and those learning the package may wish to make use of the menu system. But when you execute a command from a pull-down menu, it records the command that you could have typed in the Review window, and thus you may learn that with experience you could type that command (or modify it and resubmit it) more quickly than by use of the menus.

Stata makes reproducibility very easy through a log facility, the ability to generate a command log (containing only the commands you have entered), and the do-file editor which allows you to easily enter, execute and save sequences of commands, or program fragments.
Going one step further, if you use the do-file editor to create a sequence of commands, you may save that do-file and reuse it tomorrow, or use it as the starting point for a similar set of data management or statistical operations. Working in this way promotes reproducibility, which makes it very easy to perform an alternate analysis of a particular model. Even if many steps have been taken since the basic model was specified, it is easy to go back and produce a variation on the analysis if all the work is represented by a series of programs.

One of the implications of the concern for reproducible work: avoid altering data in a non-auditable environment such as a spreadsheet. Rather, you should transfer external data into the Stata environment as early as possible in the process of analysis, and only make permanent changes to the data with do-files that can give you an audit trail of every change made to the data.
Programmable tasks are supported by *prefix commands*, as we will soon discuss, that provide implicit loops, as well as explicit looping constructs such as the *forvalues* and *foreach* commands.

To use these commands you must understand Stata’s concepts of *local* and *global macros*. Note that the term macro in Stata bears no resemblance to the concept of an Excel macro. A macro, in Stata, is an alias to an object, which may be a number or string.
Local macros and scalars

In programming terms, *local macros* and *scalars* are the “variables” of Stata programs (not to be confused with the variables of the data set). The distinction: a local macro can contain a string, while a scalar can contain a single number (at maximum precision). You should use these constructs whenever possible to avoid creating variables with constant values merely for the storage of those constants. This is particularly important when working with large data sets.
When you want to work with a scalar object—such as a counter in a `foreach` or `forvalues` command—it will involve defining and accessing a local macro. As we will see, all Stata commands that compute results or estimates generate one or more objects to hold those items, which are saved as numeric scalars, local macros (strings or numbers) or numeric matrices.
The local macro

The *local macro* is an invaluable tool for do-file authors. A local macro is created with the *local* statement, which serves to name the macro and provide its content. When you next refer to the macro, you extract its value by *dereferencing* it, using the backtick (‘) and apostrophe (’) on its left and right. **Try it out:**

```
local george 2
local paul = ‘george’ + 2
display "‘paul’"
```

In this case, I use an equals sign in the second local statement as I want to *evaluate* the right-hand side, as an arithmetic expression, and store it in the macro `paul`. If I did not use the equals sign in this context, the macro `paul` would contain the string `2  +  2`. 
In other cases, you want to *redefine* the macro, not evaluate it, and you should not use an equals sign. You merely want to take the contents of the macro (a character string) and alter that string. The two key programming constructs for repetition, `forvalues` and `foreach`, make use of local macros as their “counter”. For instance:

```stata
forvalues i=1/10 {
    summarize PRweek`i'
}
```

Note that the value of the local macro `i` is used within the body of the loop when that counter is to be referenced. Any Stata `numlist` may appear in the `forvalues` statement. Note also the curly braces, which must appear at the end of their respective lines.
In many cases, the `forvalues` command will allow you to substitute explicit statements with a single loop construct. By modifying the range and body of the loop, you can easily rewrite your do-file to handle a different case.

The `foreach` command is even more useful. It defines an iteration over any one of a number of lists:

- the contents of a varlist (list of existing variables)
- the contents of a newlist (list of new variables)
- the contents of a numlist (list of integers)
- the separate words of a macro
- the elements of an arbitrary list
For example, we might want to **summarize** each of these variables’ detailed statistics from this World Bank data set.  

**Try it out:**

```
sysuse lifeexp
foreach v of varlist popgrowth lexp gnppc {
    summarize `v’, detail
}
```
Or, run a regression on variables for each region, and graph the data and fitted line.

**Try it out:**

```
sysuse lifeexp
levelsof region, local(regid)
foreach c of local regid {
local rr : label region `c'
    regress lexp gnppc if region ==`c'
    twoway (scatter lexp gnppc if region ==`c') ///
        (lfit lexp gnppc if region ==`c', ///
         ti(Region: `rr') name(fig`c', replace))
}
```
A local macro can be built up by redefinition.

**Try it out:**

```
sysuse lifeexp
local alleps
foreach c of local regid {
    regress lexp gnpppc if region ==`c'
predict double eps`c' if e(sample), residual
local alleps "`alleps' eps`c'"
}
```

Within the loop we redefine the macro `alleps` (as a double-quoted string) to contain itself and the name of the residuals from that region’s regression.
We could then use the macro `alleps` to generate a graph of all three regions’ residuals.

Try it out:

gen cty = _n
scatter `alleps` cty, yline(0) scheme(s2mono) legend(rows(1)) ///	ti("Residuals from model of life expectancy vs per capita GDP") ///	t2("Fit separately for each region")
Residuals from model of life expectancy vs per capita GDP

Fit separately for each region
Global macros

Stata also supports *global macros*, which are referenced by a different syntax (`$country` rather than `‘country’`). Global macros are useful when particular definitions (e.g., the default working directory for a particular project) are to be referenced in several do-files that are to be executed.

However, the creation of persistent objects of global scope can be dangerous, as global macro definitions are retained for the entire Stata session. One of the advantages of local macros is that they disappear when the do-file or ado-file in which they are defined finishes execution.
A number of Stata commands can be used as prefix commands, preceding a Stata command and modifying its behavior. The most commonly employed is the by prefix, which repeats a command over a set of categories. The statsby: prefix repeats the command, but collects statistics from each category. The rolling: prefix runs the command on moving subsets of the data (usually time series).

Several other command prefixes: simulate:, which simulates a statistical model; bootstrap:, allowing the computation of bootstrap statistics from resampled data; and jackknife:, which runs a command over jackknife subsets of the data. The svy: prefix can be used with many statistical commands to allow for survey sample design.
You can often save time and effort by using the *by* prefix. When a command is prefixed with a *bylist*, it is performed repeatedly for each element of the variable or variables in that list, each of which must be categorical.

**Try it out:**

```
sysuse census
by region: summ pop medage
```

This one command provides descriptive statistics for each of four US Census regions. If the data are not already sorted by the bylist variables, the *prefix* `bysort` should be used. The option `, total` will add the overall summary.
This can be extended to include more than one by-variable.

**Try it out:**

```bash
sysuse census
genlarge = (pop > 5000000) & !mi(pop)
bysort region large: summ popurban death
```

This is a very handy tool, which often replaces explicit loops that must be used in other programs to achieve the same end.

The by-group logic will work properly even when some of the defined groups have no observations. However, its limitation is that it can only execute a *single* command for each category. If you want to estimate a regression for each group and save the residuals or predicted values, you must use an explicit loop.
The by *prefix* should not be confused with the by *option* available on some commands, which allows for specification of a grouping variable: for instance

```
ttest price, by(foreign)
```

will run a t-test for the difference of sample means across domestic and foreign cars.

Another useful aspect of *by* is the way in which it modifies the meanings of the observation number symbol. Usually \_n refers to the current observation number, which varies from 1 to \_N, the maximum defined observation. Under a bylist, \_n refers to the observation within the bylist, and \_N to the total number of observations for that category. This is often useful in creating new variables.
For instance, if you have data on individuals with a family identifier, these commands might be useful:

```bash
sort famid age
by famid: generate famsize = _N
by famid: generate birthorder = _N - _n +1
```

Here the `famsize` variable is set to \( _N \), the total number of records for that family, while the `birthorder` variable is generated by sorting the family members’ ages within each family.
Missing values

Missing value codes in Stata appear as the dot (.) in printed output (and a string missing value code as well: “”, the null string). It takes on the largest possible positive value, so in the presence of missing data you do not want to say

```
generate hiprice = (price > 10000)
```

but rather

```
generate hiprice = (price > 10000) if price < .
```

or

```
generate hiprice = (price > 10000) if !mi(price)
```

which then generates an indicator (dummy) variable equal to 1 for high-priced cars. The indicator will be zero for low-priced cars and missing for cars with missing prices.
Stata allows for multiple missing value codes (\.a, .b, .c, ..., .z). The standard missing value code (\.) is the smallest among them, so testing for < . will always work. You may also use the missing function: \texttt{mi(varname)} will return 1 if the observation is a missing value, 0 otherwise.
An issue that often arises when importing data from external sources is the proper handling of missing data codes. Spreadsheet files often use \texttt{NA} to denote missing values, while in some datasets codes such as $-9$, $-999$, or $-0.001$ are used. The latter instances are particularly worrisome as they may not be detected unless the variables’ values are carefully scrutinized.

Note also that there is a missing value for string variables—the null, or zero-length string—which looks identical to a string of one or more space characters.
To properly handle missing values so that they are understood as such in Stata, use the `mvdecode` command. This command allows you to map various numeric values into numeric missing, or into one of the extended missing value codes `.a`, `.b`, `. . .`, `.z`

The `mvencode` command provides the inverse operation: particularly useful if you must transfer data to another package that uses some other convention for missing values.
No matter what methods you have used to input external data to the Stata workspace, you should immediately save the file in Stata format and perform the `describe` and `summarize` commands. It is much more efficient to read a Stata-format `.dta` file with `use` than to repeatedly input a text file with any of the commands discussed above. If the file is large, you may want to use the `compress` command to optimise Stata’s memory usage before saving it. `compress` is non-destructive; it never reduces the stored precision of a variable.

Before any further use is made of this datafile, examine the results of the `describe` and `summarize` commands and ensure that each variable has been input properly, and that numeric variables have sensible values for their minima and maxima.
Each variable may have its own default display format. This does not alter the contents of the variable, but only affects how it is displayed. For instance, `%9.2f` would display a two-decimal-place real number. The command

```
format varname %9.2f
```

will save that format as the default format of the variable, and

```
format date %tm
```

will format a Stata date variable into a monthly format (e.g., `1998m10`).
Variable labels

Each variable may have its own variable label. The variable label is a character string (maximum 80 characters) which describes the variable, associated with the variable via

```
label variable varname "text"
```

Variable labels, where defined, will be used to identify the variable in printed output, space permitting.
Value labels associate numeric values with character strings. They exist separately from variables, so that the same mapping of numerics to their definitions can be defined once and applied to a set of variables (e.g. 1=very satisfied...5=not satisfied may be applied to all responses to questions about consumer satisfaction). Value labels are saved in the dataset. For example:

```
label define sexlbl 0 male 1 female
label values sex sexlbl
```

The latter command associates the label `sexlbl` with the variable `sex`. Unlike other packages, Stata’s value labels are independent of variables, and the same label may be attached to any number of variables. If value labels are defined, they will be displayed in printed output instead of the numeric values.
Generating new variables

The command `generate` is used to produce new variables in the dataset, whereas `replace` must be used to revise an existing variable—and the command `replace` must always be spelled out.

A full set of functions are available for use in the `generate` command, including the standard mathematical functions, recode functions, string functions, date and time functions, and specialized functions (see `help functions` for details). Note that `generate`’s `sum()` function is a running or cumulative sum.
As mentioned earlier, `generate` operates on all observations in the current data set, producing a result or a missing value for each. You need not write explicit loops over the observations. You can, but it is usually bad programming practice to do so. You may restrict `generate` or `replace` to operate on a subset of the observations with the `if exp` or `in range` qualifiers.

The `if exp` qualifier is usually more useful, but the `in range` qualifier may be used to list a few observations of the data to examine their validity. To list observations at the end of the current data set, use `if -5/` to see the last five.
You can take advantage of the fact that the `exp` specified in `generate` may be a logical condition rather than a numeric or string value. This allows producing both the 0s and 1s of an indicator (dummy, or Boolean) variable in one command. For instance:

```stata
generate large = (pop > 5000000) & !mi(pop)
```

The condition `& !mi(pop)` makes use of two logical operators: `&`, AND, and `!`, NOT to add the qualifier that the result variable should be missing if `pop` is missing, using the `mi()` function. Although numeric functions of missing values are usually missing, creation of an indicator variable requires this additional step for safety.

The third logical operator is the Boolean OR, written as `|`. Note also that a test for equality is specified with the `==` operator (as in C). The single `=` is used only for assignment.
Keep in mind the important difference between the `$if` `$exp$` qualifier and the `$if$` (or “programmer’s if”) command. Users of some alternative software may be tempted to use a construct such as

```
generate raceid = .
if (race == "Black")
    replace raceid = 2
else if(race== "White")
    replace raceid = 3
```

which is perfectly valid syntactically. It is also useless, in that it will define the entire `raceid` variable based on the value of `race` of the first observation in the data set!
This is properly written in Stata as

```stata
generate raceid = 2 if race == "Black"
replace raceid = 3 if race == "White"
```

The `raceid` variable will be missing if `race` does not equal either of those values.
Functions for generate and replace

A number of lesser-known functions may be helpful in performing data transformations. For instance, the `inlist()` and `inrange()` functions return an indicator of whether each observation meets a certain condition: matching a value from a list or lying in a particular range.

```
generate byte newengland = ///
inlist(state, "CT", "ME", "MA", "NH", "RI", "VT")
```

```
generate byte middleage = inrange(age, 35, 49)
```

The generated variables will take a value of 1 if the condition is met and 0 if it is not. To guard against definition of missing values of `state` or `age`, add the clause `if !missing(varname)`:  

```
generate byte middleage = inrange(age, 35, 49) if !mi(age)
```
Another common data manipulation task involves extracting a part of an integer variable. For instance, firms in the US are classified by four-digit Standard Industrial Classification (SIC) codes. The first two digits represent an industrial sector. To define an industry variable from the firm’s SIC,

\[
generate \text{ ind2d} = \text{int}(\text{SIC}/100)\]

To extract the third and fourth digits, you could use

\[
generate \text{ code34} = \text{mod}(\text{SIC}, 100)\]

using the modulo function to produce the remainder.
The `cond()` function may often be used to avoid more complicated coding. It evaluates its first argument, and returns the second argument if true, the third argument if false:

```
generate endqtr = cond( mod(month, 3) == 0, ///
"Filing month", "Non-filing month")
```

Notice that in this example the `endqtr` variable need not be defined as string in the `generate` statement.
Stata contains both a `recode` command and a `recode()` function. These facilities may be used in lieu of a number of `generate` and `replace` statements. There is also a `irecode` function to create a numeric code for values of a continuous variable falling in particular brackets. For example, using a dataset containing population and median age for a number of US states:
. use census2c  
. generate size=irecode(pop, 1000, 4000, 8000, 20000)  
. label define popsize 0 "<1m" 1 "1-4m" 2 "4-8m" 3 ">8m"  
. label values size popsize  
. tabstat pop, stat(mean min max) by(size)  

Summary for variables: pop  
by categories of: size

<table>
<thead>
<tr>
<th>size</th>
<th>mean</th>
<th>min</th>
<th>max</th>
</tr>
</thead>
<tbody>
<tr>
<td></td>
<td>mean</td>
<td>min</td>
<td>max</td>
</tr>
<tr>
<td>&lt;1m</td>
<td>744.541</td>
<td>511.456</td>
<td>947.154</td>
</tr>
<tr>
<td>1-4m</td>
<td>2215.91</td>
<td>1124.66</td>
<td>3107.576</td>
</tr>
<tr>
<td>4-8m</td>
<td>5381.751</td>
<td>4075.97</td>
<td>7364.823</td>
</tr>
<tr>
<td>&gt;8m</td>
<td>12181.64</td>
<td>9262.078</td>
<td>17558.07</td>
</tr>
<tr>
<td>Total</td>
<td>5142.903</td>
<td>511.456</td>
<td>17558.07</td>
</tr>
</tbody>
</table>
Rather than categorizing a continuous variable using threshold values, we may want to group observations based on *quantiles*: quartiles, quintiles, deciles, or any other percentiles of their empirical distribution. We can readily create groupings of that sort with *xtile*:

```
. use census2c
. xtile medagequart = medage, nq(4)
. tabstat medage, stat(n mean min max) by(medagequart)
```

Summary for variables: medage

by categories of: medagequart (4 quantiles of medage)

<table>
<thead>
<tr>
<th>medagequart</th>
<th>N</th>
<th>mean</th>
<th>min</th>
<th>max</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>7</td>
<td>29.02857</td>
<td>28.3</td>
<td>29.4</td>
</tr>
<tr>
<td>2</td>
<td>4</td>
<td>29.875</td>
<td>29.7</td>
<td>30</td>
</tr>
<tr>
<td>3</td>
<td>5</td>
<td>30.54</td>
<td>30.1</td>
<td>31.2</td>
</tr>
<tr>
<td>4</td>
<td>5</td>
<td>32</td>
<td>31.8</td>
<td>32.2</td>
</tr>
</tbody>
</table>

Total   | 21  | 30.25714| 28.3| 32.2  |
A problem that commonly arises with data transferred from spreadsheets is the automatic classification of a variable as string rather than numeric. This often happens if the first value of such a variable is \texttt{NA}, denoting a missing value. If Stata’s convention for numeric missings—the dot, or full stop (. ) is used, this will not occur. If one or more variables are misclassified as string, how can they be modified?

First, a warning. Do not try to maintain long numeric codes (such as US Social Security numbers, with nine digits) in numeric form, as they will generally be rounded off. Treat them as string variables, which may contain up to 2,045 bytes.
If a variable has merely been misclassified as string, the brute-force approach can be used:

```stata
generate patid = real( patientid )
```

Any values of `patientid` that cannot be interpreted as numeric will be missing in `patid`. Note that this will also occur if numbers are stored with commas separating thousands.
A more subtle approach is given by the `destring` command, which can transform variables in place (with the `replace` option) and can be used with a `varlist` to apply the same transformation to a set of variables. Like the `real()` function, `destring` should only be used on variables misclassified as strings.

If the variable truly has string content and you need a numeric equivalent, for statistical analysis, you may use `encode` on the variable. To illustrate, let us read in some tab-delimited data with `import delimited`.
. import delimited statedata, clear
(4 vars, 7 obs)
. format pop2008 %7.3f
. list, sep(0)

<table>
<thead>
<tr>
<th>state</th>
<th>abbrev</th>
<th>yearjo_d</th>
<th>pop2008</th>
</tr>
</thead>
<tbody>
<tr>
<td>Massachusetts</td>
<td>MA</td>
<td>1788</td>
<td>6.498</td>
</tr>
<tr>
<td>New Hampshire</td>
<td>NH</td>
<td>1788</td>
<td>1.316</td>
</tr>
<tr>
<td>Vermont</td>
<td>VT</td>
<td>1791</td>
<td>0.621</td>
</tr>
<tr>
<td>New Jersey</td>
<td>NJ</td>
<td>1787</td>
<td>8.683</td>
</tr>
<tr>
<td>Michigan</td>
<td>MI</td>
<td>1837</td>
<td>10.003</td>
</tr>
<tr>
<td>Arizona</td>
<td>AZ</td>
<td>1912</td>
<td>6.500</td>
</tr>
<tr>
<td>Alaska</td>
<td>AK</td>
<td>1959</td>
<td>0.686</td>
</tr>
</tbody>
</table>

As the data are tab-delimited, I can read a file with embedded spaces in the state variable.
I want to create a categorical variable identifying each state with an (arbitrary) numeric code. This can be achieved with `encode`:

```
. encode state, gen(stid)
. list state stid, sep(0)
```

<table>
<thead>
<tr>
<th>state</th>
<th>stid</th>
</tr>
</thead>
<tbody>
<tr>
<td>Massachusetts</td>
<td>Massachusetts</td>
</tr>
<tr>
<td>New Hampshire</td>
<td>New Hampshire</td>
</tr>
<tr>
<td>Vermont</td>
<td>Vermont</td>
</tr>
<tr>
<td>New Jersey</td>
<td>New Jersey</td>
</tr>
<tr>
<td>Michigan</td>
<td>Michigan</td>
</tr>
<tr>
<td>Arizona</td>
<td>Arizona</td>
</tr>
<tr>
<td>Alaska</td>
<td>Alaska</td>
</tr>
</tbody>
</table>

```
. summarize stid
```

<table>
<thead>
<tr>
<th>Variable</th>
<th>Obs</th>
<th>Mean</th>
<th>Std. Dev.</th>
<th>Min</th>
<th>Max</th>
</tr>
</thead>
<tbody>
<tr>
<td>stid</td>
<td>7</td>
<td>4</td>
<td>2.160247</td>
<td>1</td>
<td>7</td>
</tr>
</tbody>
</table>
Although `stid` is a numeric variable (as `summarize` shows) it is automatically assigned a *value label* consisting of the contents of `state`. The variable `stid` may now be used in analyses requiring numeric variables.

You may also want to make a variable into a string (for instance, to reinstate leading zeros in an id code variable). You may use the `string()` function, the `tostring` command or the `decode` command to perform this operation.
Stata is not limited to using the set of defined `generate` functions. The `egen` (extended `generate`) command makes use of functions written in the Stata ado-file language, so that `_gzap.ado` would define the extended generate function `zap()`. This would then be invoked as

```
egen newvar = zap(oldvar)
```

which would do whatever `zap` does on the contents of `oldvar`, creating the new variable `newvar`.

A number of `egen` functions provide row-wise operations similar to those available in a spreadsheet: row sum, row average, row standard deviation, and so on. Users may write their own `egen` functions. In particular, `findit egenmore` for a very useful collection.
Although the syntax of an *egen* statement is very similar to that of *generate*, several differences should be noted. As only a subset of *egen* functions allow a *by* *varlist*: prefix or *by*(*varlist*) option, the documentation should be consulted to determine whether a particular function is *byable*, in Stata parlance. Similarly, the explicit use of _n and _N, often useful in *generate* and *replace* commands is not compatible with *egen*. 
Wildcards may be used in row-wise functions. If you have state-level U.S. Census variables `pop1890, pop1900, ..., pop2000` you may use `egen nrCensus = rowmean(pop*)` to compute the average population of each state over those decennial censuses. The row-wise functions operate in the presence of missing values. The mean will be computed for all 50 states, although several were not part of the US in 1890.
The number of non-missing elements in the row-wise `varlist` may be computed with `rownonmiss()`, with `rowmiss()` as the complementary value. Other official row-wise functions include `rowmax()`, `rowmin()`, `rowtotal()` and `rowsd()` (row standard deviation).

The functions `rowfirst()` and `rowlast()` give the first (last) non-missing values in the `varlist`. You may find this useful if the variables refer to sequential items: for instance, wages earned per year over several years, with missing values when unemployed. `rowfirst()` would return the earliest wage observation, and `rowlast()` the most recent.
Official *egen* also provides a number of statistical functions which compute a statistic for specified observations of a variable and place that constant value in each observation of the new variable. Since these functions generally allow the use of `by varlist:`, they may be used to compute statistics for each *by-group* of the data. This facilitates computing statistics for each household for individual-level data or each industry for firm-level data. The `count()`, `mean()`, `min()`, `max()` and `total()` functions are especially useful in this context.

As an illustration using our state-level data, we *egen* the average population in each of the *size* groups defined above, and express each state’s population as a percentage of the average population in that size group. Size category 0 includes the smallest states in our sample.
. use census2c
. bysort size: egen avgpop = mean(pop)
. generate popratio = 100 * pop / avgpop
. format popratio %7.2f
. list state pop avgpop popratio if size == 0, sep(0)

<table>
<thead>
<tr>
<th>state</th>
<th>pop</th>
<th>avgpop</th>
<th>popratio</th>
</tr>
</thead>
<tbody>
<tr>
<td>1. Rhode Island</td>
<td>947.2</td>
<td>744.541</td>
<td>127.21</td>
</tr>
<tr>
<td>2. Vermont</td>
<td>511.5</td>
<td>744.541</td>
<td>68.69</td>
</tr>
<tr>
<td>3. N. Dakota</td>
<td>652.7</td>
<td>744.541</td>
<td>87.67</td>
</tr>
<tr>
<td>4. S. Dakota</td>
<td>690.8</td>
<td>744.541</td>
<td>92.78</td>
</tr>
<tr>
<td>5. New Hampshire</td>
<td>920.6</td>
<td>744.541</td>
<td>123.65</td>
</tr>
</tbody>
</table>
Other `egen` functions in this statistical category include `iqr()` (inter-quartile range), `kurt()` (kurtosis), `mad()` (median absolute deviation), `mdev()` (mean absolute deviation), `median()`, `mode()`, `pc()` (percent or proportion of total), `pctile()`, `p(n)` (\(n^{th}\) percentile), `rank()`, `sd()` (standard deviation), `skew()` (skewness) and `std()` (z-score).

Many other `egen` functions are available; see `help egen` for details.
Time series calendar

Stata supports date (and time) variables and the creation of a time series calendar variable. Dates are expressed, as they are in Excel, as the number of days from a base date. In Stata’s case, that date is 1 Jan 1960 (like Unix/Linux). You may set up data on an annual, half-yearly, quarterly, monthly, weekly or daily calendar, as well as a calendar that merely uses the observation number.

You may also set the `delta` of the calendar variable to be other than 1: for instance, if you have data at five-year intervals, you may define the data as annual with `delta=5`. This ensures that the lagged value of the 2005 observation is that of 2000.
A useful utility for setting the appropriate time series calendar is `tsmktim`, available from the SSC Archive (`ssc describe tsmktim`) and described in “Utility for time series data”, Baum, CF and Wiggins, VL. *Stata Technical Bulletin*, 2000, 57, 2-4. It will set the calendar, issuing the appropriate `tsset` command and the display format of the resulting calendar variable, and can be used in a panel data context where each time series starts in the same calendar period.
An observation-number calendar is generally necessary for business-daily data where you want to avoid gaps for weekends, holidays etc. which will cause lagged values and differences to contain missing values. However, you may want to create two calendar variables for the same time series data: one for statistical purposes and one for graphical purposes, which will allow the series to be graphed with calendar-date labels. This procedure is illustrated in “Stata Tip 40: Taking care of business...”, Baum, CF. *Stata Journal*, 2007, 7:1, 137-139.

This is a moot point in Stata versions 12 and 13, which provide support for custom business-daily calendars (or *bcal*). As we shall see, Stata can construct the *bcal* from your dataset in version 13.
The $D$, $L$, and $F$ operators may be used under a time series calendar (including in the context of panel data) to specify first differences, lags, and leads, respectively. These operators understand missing data, and numlists: e.g. $L(1/4).x$ is the first through fourth lags of $x$, while $L2D.x$ is the second lag of the first difference of the $x$ variable.
It is very important to use the time series operators to refer to lagged or led values, rather than referring to the observation number (e.g., \(_n-1\)). The time series operators respect the time series calendar, and will not mistakenly compute a lag or difference from a prior period if it is missing. This is particularly important when working with panel data to ensure that references to one individual do not reach back into the prior individual’s data.
Using time series operators, you may not only consistently generate differences, lags, and leads, but may refer to them ‘on the fly’ in statistical and estimation commands. To estimate an AR(4) model, you need not create the lagged variables.

**Try it out!**

```
webuse lutkepohl, clear
regress consumption L(1/4).consumption
```

To test Granger causality:

```
regress consumption L(-4/4).income
```

which would regress consumption on four leads, four lags and the current value of income.

For a “Dickey–Fuller” style regression:

```
regress D.consumption L.consumption
```
Comma-separated (CSV) files or tab-delimited data files may be read very easily with the `import delimited` command. If your file has variable names in the first row that are valid for Stata, they will be automatically used (rather than default variable names). You usually need not specify whether the data are tab- or comma-delimited. Note that `import delimited` cannot read space-delimited data, nor character strings with embedded spaces, unless they are tab-delimited.
The command presumes that the filetype is `.csv`:

```stata
import delimited using filename
```

You can also specify which rows and/or columns of the `.csv` file are to be read.
In Stata 12 and 13, you may read data directly from Excel and Excel-compatible worksheets with the `import excel` command. You may specify from which sheet the data are to be loaded, and the range of that sheet in which they are located. You may also specify that the first row contains valid Stata variable names that are to be used. For example,

```
import excel using weo_201204_FR.xls, firstrow clear
```
We may read only some of the columns, but to use the Stata variable names rather than column letters, we must specify `cellrange()` to omit the first row rather than the `firstrow` option. For example,

import excel using weo_201204_FR.xls, describe

import excel iso year NGDPPC PCPI using weo_201204_FR.xls, cellrange(A2:AW39) clear
A free-format ASCII text file with space-, tab-, or comma-delimited data may be read with the `infile` command. The missing-data indicator (.) may be used to specify that values are missing.

The command must specify the variable names. Assuming `auto.raw` contains numeric data,

```
infile price mpg displacement using auto
```

will read it. If a file contains a combination of string and numeric values in a variable, it should be read as string, and `encode` used to convert it to numeric with string value labels.
If some of the data are string variables without embedded spaces, they must be specified in the command:

```
infile str3 country price mpg displacement using auto2
```

would read a three-letter country of origin code, followed by the numeric variables. The number of observations will be determined from the available data.
The `infile` command may also be used with fixed-format data, including data containing undelimited string variables, by creating a dictionary file which describes the format of each variable and specifies where the data are to be found. The dictionary may also specify that more than one record in the input file corresponds to a single observation in the data set.

Sometimes data fields are not delimited—for instance, the sequence ‘102’ might actually represent three integer variables. A dictionary must then be used to define the variables’ locations.

The `byvariable()` option allows a variable-wise dataset to be read, where one specifies the number of observations available for each series.
If your data are already in the internal format of SAS, SPSS, Excel, GAUSS, MATLAB, or a number of other packages, the best way to get it into Stata is by using the third-party product Stat/Transfer.

Stat/Transfer will preserve variable labels, value labels, and other aspects of the data, and can be used to convert a Stata binary file into other packages’ formats. It can also produce subsets of the data (selecting variables, cases or both) so as to generate an extract file that is more manageable. This is particularly important when the 2,047-variable limit on standard Stata data sets is encountered.
If you want to transfer data to another package, Stat/Transfer is very useful. But if you just want to create an ASCII file from Stata, the `outfile` command may be used. It takes a `varlist`, and the `if` or `in` clauses may be used to control the observations to be exported. Applying `sort` prior to `outfile` will control the order of observations in the external file. You may specify that the data are to be written in comma-separated format.
The `export delimited` command can write a comma-delimited or tab-delimited ASCII file, optionally placing the variable names in the first row. Such a file can be easily read by a spreadsheet program such as Excel.

For customized output, the `file` command can write out information (including scalars, matrices and macros, text strings, etc.) in any ASCII or binary format of your choosing.
In Stata 12 and 13, the `export excel` command may be used to write into a sheet of an Excel or Excel-compatible workbook. You may modify or replace an existing sheet, and specify which Stata variables are to be written to the worksheet. For example, after importing the French spreadsheet data:

```plaintext
export excel using myfrenchdata, firstrow(variables) replace
export excel iso year NGDPPC PCPI using myfrenchdata, replace
```

will create a new workbook, `myfrenchdata.xls`, containing all variables in memory in the first example, and only the four specified variables in the second example. You may also specify that Stata variable labels are to be output, rather than variable names.
A very useful capability is provided by the `postfile` and `post` commands, which permit a Stata data set to be created in the course of a program. For instance, you may be simulating the distribution of a statistic, fitting a model over separate samples, or bootstrapping standard errors. Within the looping structure, you may `post` certain numeric values to the `postfile`. This will create a separate Stata binary data set, which may then be opened in a later Stata run and analyzed. Note that the parens () given in the documentation, surrounding each `exp`, are required.
In many empirical research projects, the raw data to be utilized are stored in a number of separate files: separate “waves” of panel data, timeseries data extracted from different databases, and the like. Stata only permits a single data set to be accessed at one time. How, then, do you work with multiple data sets? Several commands are available, including `append`, `merge`, and `joinby`.

How, then, do you combine datasets in Stata? First of all, it is important to understand that at least one of the datasets to be combined must already have been saved in Stata format. Second, you should realize that each of Stata’s commands for combining datasets provides a certain functionality, which should not be confused with that of other commands.
The **append** command combines two Stata-format data sets that possess variables in common, adding observations to the existing variables. The same variables need not be present in both files, as long as a subset of the variables are common to the “master” and “using” data sets. It is important to note that “PRICE” and “price” are different variables, and one will not be appended to the other.
You might have a dataset on the demographic characteristics in 2007 of the largest municipalities in China, `cityCN`. If you were given a second dataset containing the same variables for the largest municipalities in Japan in 2007, `cityJP`, you might want to combine those datasets with `append`. With the `cityCN` dataset in memory, you would `append using cityJP`, which would add those records as additional observations. You could then save the combined file under a different name. `append` can be used to combine multiple datasets, so if you had the additional files `cityPH` and `cityMY`, you could list those filenames in the `using` clause as well.

Prior to using `append`, it is a good idea to create an identifier variable in each dataset that takes on a constant value: e.g., `gen country = 1` in the CN dataset, `gen country = 2` in the JP dataset, etc.
For instance, consider the `append` command with two stylized datasets:

**dataset1**:  
\[
\begin{pmatrix}
id & var1 & var2 \\
112 & \cdot & \cdot \\
216 & \cdot & \cdot \\
449 & \cdot & \cdot \\
\end{pmatrix}
\]

**dataset2**:  
\[
\begin{pmatrix}
id & var1 & var2 \\
126 & \cdot & \cdot \\
309 & \cdot & \cdot \\
421 & \cdot & \cdot \\
604 & \cdot & \cdot \\
\end{pmatrix}
\]
These two datasets contain the same variables, as they must for `append` to sensibly combine them. If `dataset2` contained `idcode, Var1, Var2` the two datasets could not sensibly be appended without renaming the variables (recall that in Stata, `var1` and `Var1` are two separate variables). Appending these two datasets with common variable names creates a single dataset containing all of the observations:
The rule for \texttt{append}, then, is that if datasets are to be combined, they should share the same variable names and datatypes (string vs. numeric). In the above example, if \texttt{var1} in \texttt{dataset1} was a float while that variable in \texttt{dataset2} was a string variable, they could not be appended.
It is permissible to append two datasets with differing variable names in the sense that `dataset2` could also contain an additional variable or variables (for example, `var3`, `var4`). The values of those variables in the observations coming from `dataset1` would then be set to missing.

Some care must be taken when appending datasets in which the same variable may exist with different data types (string in one, numeric in another). For details, see “Stata tip 73: append with care!”, Baum CF, *Stata Journal*, 2008, 9:1, 166-168.
We now describe the `merge` command, which is Stata’s basic tool for working with more than one dataset. Its syntax changed considerably in Stata version 11.

The merge command takes a first argument indicating whether you are performing a `one-to-one`, `many-to-one`, `one-to-many` or `many-to-many` merge using specified key variables. It can also perform a one-to-one merge by observation.
Like the `append` command, the `merge` works on a “master” dataset—the current contents of memory—and a single “using” dataset (prior to Stata 11, you could specify multiple using datasets). One or more key variables are specified, and you need not sort either dataset prior to merging.

The distinction between “master” and “using” is important. When the same variable is present in each of the files, Stata’s default behavior is to hold the master data inviolate and discard the using dataset’s copy of that variable. This may be modified by the `update` option, which specifies that non-missing values in the using dataset should replace missing values in the master, and the even stronger `update replace`, which specifies that non-missing values in the using dataset should take precedence.
A “one-to-one” merge (written `merge 1:1`) specifies that each record in the using data set is to be combined with one record in the master data set. This would be appropriate if you acquired additional variables for the same observations.

In any use of `merge`, a new variable, `_merge`, takes on integer values indicating whether an observation appears in the master only, the using only, or appears in both. This may be used to determine whether the merge has been successful, or to remove those observations which remain unmatched (e.g. merging a set of households from different cities with a comprehensive list of postal codes; one would then discard all the unused postal code records). The `_merge` variable must be dropped before another `merge` is performed on this data set.
Consider these two stylized datasets:

<table>
<thead>
<tr>
<th>id</th>
<th>var1</th>
<th>var2</th>
</tr>
</thead>
<tbody>
<tr>
<td>112</td>
<td></td>
<td></td>
</tr>
<tr>
<td>216</td>
<td></td>
<td></td>
</tr>
<tr>
<td>449</td>
<td></td>
<td></td>
</tr>
</tbody>
</table>

<table>
<thead>
<tr>
<th>id</th>
<th>var22</th>
<th>var44</th>
<th>var46</th>
</tr>
</thead>
<tbody>
<tr>
<td>112</td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>216</td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>449</td>
<td></td>
<td></td>
<td></td>
</tr>
</tbody>
</table>

*dataset1*:

*dataset3*:
We may merge these datasets on the common `merge key`: in this case, the `id` variable.

\[
\begin{pmatrix}
\text{id} & \text{var1} & \text{var2} & \text{var22} & \text{var44} & \text{var46} \\
112 & : & : & : & : & : \\
216 & : & : & : & : & : \\
449 & : & : & : & : & : \\
\end{pmatrix}
\]
The rule for `merge`, then, is that if datasets are to be combined on one or more `merge keys`, they each must have one or more variables with a common name and datatype (string vs. numeric). In the example above, each dataset must have a variable named `id`. That variable can be numeric or string, but that characteristic of the merge key variables must match across the datasets to be merged. Of course, we need not have exactly the same observations in each dataset: if `dataset3` contained observations with additional `id` values, those observations would be merged with missing values for `var1` and `var2`.

This is the simplest kind of merge: the `one-to-one merge`. Stata supports several other types of merges. But the key concept should be clear: the `merge` command combines datasets “horizontally”, adding variables’ values to existing observations.
The `merge` command can also do a “many-to-one” or “one-to-many” merge. For instance, you might have a dataset named `hospitals` and a dataset named `discharges`, both of which contain a hospital ID variable `hospid`. If you had the `hospitals` dataset in memory, you could `merge 1:m hospid using discharges` to match each hospital with its prior patients. If you had the `discharges` dataset in memory, you could `merge m:1 hospid using hospitals` to add the hospital characteristics to each discharge record. This is a very useful technique to combine aggregate data with disaggregate data without dealing with the details.

Although “many-to-one” or “one-to-many” merges are commonplace and very useful, you should never want to do a “many-to-many” (m:m) merge, which will yield seemingly random results.
The long-form dataset is very useful if you want to add aggregate-level information to individual records. For instance, we may have panel data for a number of companies for several years. We may want to attach various macro indicators (interest rate, GDP growth rate, etc.) that vary by year but not by company. We would place those macro variables into a dataset, indexed by year, and sort it by year.

We could then use the firm-level panel dataset and sort it by year. A merge command can then add the appropriate macro variables to each instance of year. This use of merge is known as a one-to-many match merge, where the year variable is the merge key.

Note that the merge key may contain several variables: we might have information specific to industry and year that should be merged onto each firm’s observations.
Data are often provided in a different orientation than that required for statistical analysis. The most common example of this occurs with panel, or longitudinal, data, in which each observation conceptually has both cross-section \((i)\) and time-series \((t)\) subscripts. Often one will want to work with a “pure” cross-section or “pure” time-series. If the microdata themselves are the objects of analysis, this can be handled with sorting and a loop structure. If you have data for \(N\) firms for \(T\) periods per firm, and want to fit the same model to each firm, one could use the `statsby` command, or if more complex processing of each model’s results was required, a `foreach` block could be used. If analysis of a cross-section was desired, a `bysort` would do the job.
But what if you want to use average values for each time period, averaged over firms? The resulting dataset of $T$ observations can be easily created by the `collapse` command, which permits you to generate a new data set comprised of summary statistics of specified variables. More than one summary statistic can be generated per input variable, so that both the number of firms per period and the average return on assets could be generated. `collapse` can produce counts, means, medians, percentiles, extrema, and standard deviations.
Different models applied to longitudinal data require different orientations of those data. For instance, seemingly unrelated regressions \((\text{sureg})\) require the data to have \(T\) observations ("wide"), with separate variables for each cross-sectional unit. Fixed-effects or random-effects regression models \(\text{xtreg}\), on the other hand, require that the data be stacked or "vec"d in the "long" format. It is usually much easier to generate transformations of the data in stacked format, where a single variable is involved.

The \texttt{reshape} command allows you to transfer the data from the former ("wide") format to the latter ("long") format or vice versa. It is a complicated command, because of the many variations on this process one might encounter, but it is very powerful.
When data have more than one identifier per record, they may be organized in different ways. For instance, it is common to find on-line displays or downloadable spreadsheets of data for individual units—for instance, U.S. states—with the unit’s name labeling the row and the year labeling the column. If these data were brought into Stata in this form, they would be in the **wide form**, wide form with the same measurement (population) for different years denoted as separate Stata variables:

```stata
.list, noobs
```

<table>
<thead>
<tr>
<th>state</th>
<th>pop1990</th>
<th>pop1995</th>
<th>pop2000</th>
</tr>
</thead>
<tbody>
<tr>
<td>CT</td>
<td>3291967</td>
<td>3324144</td>
<td>3411750</td>
</tr>
<tr>
<td>MA</td>
<td>6022639</td>
<td>6141445</td>
<td>6362076</td>
</tr>
<tr>
<td>RI</td>
<td>1005995</td>
<td>1017002</td>
<td>1050664</td>
</tr>
</tbody>
</table>
There are a number of Stata commands—such as `egen` row-wise functions—which work effectively on data stored in the wide form. It may also be a useful form of data organization for producing graphs.

Alternatively, we can imagine stacking each year’s population figures from this display into one variable, `pop`. In this format, known in Stata as the *long form*, each datum is identified by two variables: the state name and the year to which it pertains.
We use `reshape` to transform the data, indicating that `state` should be the main row identifier (i) with `year` as the secondary identifier (j):

```
. reshape long pop, i(state) j(year)
```

```
. list, noobs sepby(state)
```

<table>
<thead>
<tr>
<th>state</th>
<th>year</th>
<th>pop</th>
</tr>
</thead>
<tbody>
<tr>
<td>CT</td>
<td>1990</td>
<td>3291967</td>
</tr>
<tr>
<td>CT</td>
<td>1995</td>
<td>3324144</td>
</tr>
<tr>
<td>CT</td>
<td>2000</td>
<td>3411750</td>
</tr>
<tr>
<td>MA</td>
<td>1990</td>
<td>6022639</td>
</tr>
<tr>
<td>MA</td>
<td>1995</td>
<td>6141445</td>
</tr>
<tr>
<td>MA</td>
<td>2000</td>
<td>6362076</td>
</tr>
<tr>
<td>RI</td>
<td>1990</td>
<td>1005995</td>
</tr>
<tr>
<td>RI</td>
<td>1995</td>
<td>1017002</td>
</tr>
<tr>
<td>RI</td>
<td>2000</td>
<td>1050664</td>
</tr>
</tbody>
</table>
This data structure is required for many of Stata’s statistical commands, such as the `xt` suite of panel data commands. The long form is also very useful for data management using `by`-groups and the computation of statistics at the individual level, often implemented with the `collapse` command.

Inevitably, you will acquire data (either raw data or Stata datasets) that are stored in either the wide or the long form and will find that translation to the other format is necessary to carry out your analysis. In statistical packages lacking a data-reshape feature, common practice entails writing the data to one or more external text files and reading it back in.
With the proper use of `reshape`, writing data out and reading them back in is not necessary in Stata. But `reshape` requires, first of all, that the data to be reshaped are labelled in such a way that they can be handled by the mechanical rules that the command applies. In situations beyond the simple application of `reshape`, it may require some experimentation to construct the appropriate command syntax. This is all the more reason for enshrining that code in a do-file as some day you are likely to come upon a similar application for `reshape`.